20/16-bit AGC Architecture

Harvard architecture with a ROM of 32k words of 20 bits for instructions only (6 μsec); a RAM of up to 4k words of 16 bits for data only (6 μsec); 1 parity bit in ROM & RAM.

Central registers of 16 bits with no parity bit: A & L, ALS, Q, Z, X, Y, B, RAP.

Other central registers with no parity bit: IR/14, SF[ixed]/15, SE[rasable]/12.

Input/Output registers or channels TBD.

No program interrupt logic; instead, events and timeouts set bits for RUPT to test.

A – Accumulator for single precision and upper half of double precision.
L – Lower accumulator for double precision, specifically including lower half of a product from MP and MPAD, and lower half of a dividend for DV.

ALS – A&L Shadow, used in shifting A and L together, also as AS and LS.

Q – Quick-turnover register for return address and other very-short-term data.

Only the above central registers are addressable, as 0000, 0001, and 0002.

Z – Instruction location counter, usually contains the (necessarily ROM) address of the next instruction; see also incrementing  property of SF.
X – First latch for augend in any addition except Z increment.
Y – Second latch for addend; pseudo ‘U’ is the sum.
B – Buffer for multiple purposes; pseudo ‘C’ is its ones-complement outputs.
RAP – Return-Address Pointer, for LIFO handling of RAs in low RAM.

IR – Interrupt Requests, prioritized array of bits, one per interrupt type.

SF – Select Fixed: holds address to drive ROM; incremented content is readable..
SE – Select Erasable: holds address to drive RAM.
General Description

Instructions of 19 bits plus odd parity occupy up to 32K of ROM (aka Fixed). Data words of 15 bits plus odd parity occupy up to 4K words of RAM (aka Erasable). A few central registers (A, L, Q) have low-number addresses to be addressable like RAM. Some low-number RAM locations are special in the sense that their content is modified by external events, most often incrementing or decrementing to track spacecraft sensor values. There are no special-register locations that perform shifting or other data editing. Whenever possible, a Memory Cycle Time (MCT) performs RAM and ROM cycles simultaneously.
There are 8 instructions that have 15-bit address fields, of which two are special cases because their 15-bit field is constant data which one emits into the A register, and the other substitutes it for whatever data is next read from erasable. The others use their 15-bit fields to address ROM (only) for transfer-of-control purposes.
Of the other instructions, none can address ROM. They all. Those that need to address data in RAM devote 12 bits to the purpose, and include a 13th bit to indicate indirect addressing. When they perform indirect addressing, the final address is taken from the rightmost 12 bit positions, ignoring the leftmost 3 bits which can be used for loop counting. There is one exceptional case of indirect addressing, PRET*, which has no direct-addressing form, and uses the indirect address to access a word of interpretive code in ROM. There is an INDEX instruction that adds part or all of its 15-bit operand to the 19-bit next instruction, not affecting any bits above the address field defined for that instruction type. Instructions addressing A, L, or Q access RAM when appropriate.
A comprehensive set of shifting instructions, all double precision, support the native add and subtract instructions in both precisions. Multiplication uses single-precision inputs to create a double precision product, and has an option to include the addition of a single-precision number aligned with the lower half of the product. Division uses a double-precision dividend and a single-precision divisor, yielding a single-precision quotient and a single-precision remainder.

Input and output instructions address numbered channels and can perform Boolean functions to read or write only selected bits in any channel. Also, a variety of I/O operations work by stealing an individual RAM cycle, usually to increment or decrement a data word in RAM.

There are a number of 19-bit codes corresponding to miscellaneous operations such as absolute value, normalizing shift, square root, and sine-cosine. An important example is REVIF, which reverses the sense of the next IF type. Of these, as many as possible are implemented within the budget of logic hardware; the remainder call emulation routines. Having no address field, these are unaffected by INDEX.
There is no program interrupt in the customary sense. Instead, any timing out or other event that can change the priority of a job sets a flag bit in an Interrupt Request register, and at convenient but frequent moments the program is required to perform the no-address RUPT instruction (take highest priority interRUPT). If it finds any non-zero bits, it copies the IR word into A, puts its own location into Q, and transfers control to ROM location 00000. The hardware restart GOJAM clears all central registers, thus transferring control to 00000, where software recognizes that c(A) = 0 implies GOJAM.
The go-to and call functions are performed by the same operation code, documented as CALL to gratify Dijkstra fans, even though the assembler accepts GOTO or CALL. Any of the five IF instructions also functions as CALL if its branching condition is satisfied, copying Z into Q to serve as a return address. There is an EXEC instruction that gives control to the addressed instruction without changing Z, so that any instruction that doesn’t arbitrarily change Z performs its function and uses the unchanged Z to pass control back to whatever follows the EXEC. The exceptions are CALL and any IF-type instruction that actually branches. A special execute type, PRET*, prepares its ROM operand for processing as an interpretive instruction by distributing its 19 bits to A and L.
Parts of the design are motivated by the need, in abnormal circumstances, to insert into RAM something that will act like a program even though it can’t quite look like one. Ground controllers use a piece of support software that breaks each 19-bit instruction created on the ground into two 15-bit words to insert into available RAM space. They then direct the computer to a program that uses these parameters to effectively interpret the machine’s own native instructions by executing a standard copy of each desired instruction that generally uses indirect addressing to access the desired location.
Instruction Formats in ROM

Immediate data and instruction reference types (branches):

D...D is immediate data; A…A is address, ROM only. Both types are indexable.
Octl Op Code 19 18 17 16 15 14 13 12 11 10 09 08 07 06 05 04 03 02 01

1 0  DATA=   1  0  0  0  D  D  D  D  D  D  D  D  D  D  D  D  D  D  D

1 1  DATAA   1  0  0  1  D  D  D  D  D  D  D  D  D  D  D  D  D  D  D

1 2  EXEC    1  0  1  0  A  A  A  A  A  A  A  A  A  A  A  A  A  A  A

1 3  CALL    1  0  1  1  A  A  A  A  A  A  A  A  A  A  A  A  A  A  A
1 4  IFPNZ   1  1  0  0  A  A  A  A  A  A  A  A  A  A  A  A  A  A  A

1 5  IFNZ    1  1  0  1  A  A  A  A  A  A  A  A  A  A  A  A  A  A  A
1 6  IFNEG   1  1  1  0  A  A  A  A  A  A  A  A  A  A  A  A  A  A  A
1 7  IFNOV   1  1  1  1  A  A  A  A  A  A  A  A  A  A  A  A  A  A  A

Data handling operations that use RAM addresses for direct or indirect (*):

Op Code 19 18 17 16 15 14 13 12 11 10 09 08 07 06 05 04 03 02 01

(gen’l) 0  C  C  C  C  C  *  A  A  A  A  A  A  A  A  A  A  A  A

0 00 – 0 06  LDA & LDL & DLD & LDQ   4  4

0 10 – 0 16  STA & STL & DST & STQ   4  8

0 20 – 0 26  XCA & XCL & DXC & XCQ   4 12

0 30 – 0 36  T   & DT  & CP  & DCP   4 16

0 40 – 0 46  AD  & DAD & SU  & DSU   4 20
0 50 – 0 56  MP  & MSK & DV  & MPAD  4 24
0 60 – 0 62  LAN & DLN               2 26
0 64 – 0 70  INC & RED & NDX         3 29
0 72 – 0 73  STNA(direct) & PRET*    1 30
0 74 – 0 75  shift & inout (no *)    1 31

0 76 – 0 77  miscellaneous (no *)    1 32

The LD/ST/XC instructions have special modes when directly addressing the central registers they serve: LD→T [Test], ST→Z [clear to Zero], XC→C [Complement], e.g. TA = LDA A, ZL = STL L, CQ = XCQ Q.
Indirect address used by all but the last 3 lines of this type; LLL is a loop counter (1-8):
15 14 13 12 11 10 09 08 07 06 05 04 03 02 01
L  L  L  E  E  E  E  E  E  E  E  E  E  E  E [Addresses RAM only]
Loop counter LLL is used only by RED, which decrements E…E by 1 or 2 per last op and tests LLL for 000, setting condition code; decrements LLL by 1 if it was not 000. PRET* is always indirect and always accesses ROM through its indirect address.
Op Code 19 18 17 16 15 14 13 12 11 10 09 08 07 06 05 04 03 02 01

(shift) 0  1  1  1  1  0  0  P  P  U  D  S  S  S  N  N  N  N  N

All shifts affect both A and L, but PP can clear either A or L initially; U=unsigned; DSSSNNNNN=twos-complement count with extended sign bit to determine direction: positive=left, negative=right. INDEX affects bits 09-01 only. 

Op Code 19 18 17 16 15 14 13 12 11 10 09 08 07 06 05 04 03 02 01

(inout) 0  1  1  1  1  0  1  D  B  B  C  C  C  C  C  C  C  C  C
D=direction, BB=Boolean function, C…C=channel. INDEX affects bits 09-01 only.
Op Code 19 18 17 16 15 14 13 12 11 10 09 08 07 06 05 04 03 02 01

(misc)  0  1  1  1  1  1  E  C  C  C  C  C  C  C  C  C  C  C  C

Miscellaneous: RUPT, SQRT, etc. E=Emulated rather than native. C…C=detail code or call address in low ROM.
Brief Explanation of Instruction Operation Codes

Immediate data and instruction reference (branching); none of these fetches DATA=
DATA=
General immediate DATA, for use with any instruction that fetches data.
 

Bits 15-01 replace whatever is next read from RAM, on a one-shot basis.
DATAA
Default case for loading immediate DATA into A.

EXEC

EXECutes addressed instruction without transferring control.

CALL

CALL or GOTO, unconditional; target decides which (also true of IFs).

IFPNZ

IF condition code shows Positive Non-Zero, branch.

IFNZ

IF condition code shows Non-Zero, branch.

IFNEG

IF condition code shows NEGative (necessarily non-zero), branch.

IFNOV
IF condition code shows NO oVerflow, branch.

The sense of any IF is reversed if it immediately follows a REVIF instruction:
IFN|Z

IF condition code shows Negative or Zero, branch.

IFZER

IF condition code shows ZERo, branch.

IFP|Z

IF condition code shows Positive or Zero, branch.

IFOVF

IF condition code shows OVerFlow, branch.

Operations that use erasable (RAM) for fetching and/or storing:

LDA

LoaD A (LDA A = TA: Test A[?])
LDL

LoaD L (LDL L = TL: Test L[?])

DLD

Double LoaD into A and L (DLD L = TD: Test Double c(A,L[?])

LDQ

LoaD Q (LDQ Q = TQ: Test Q[?])

STA

STore A (STA A = ZA: Zero A)

STL

STore L (STL L = ZL: Zero L)

DST

Double Store from A and L (DST L = ZD: Zero Double c(A,L)

STQ

STore Q (STQ Q = ZQ: Zero Q)

XCA

eXChange A (XCA A = CA: Complement A)

XCL

eXChange L (XCL L = CL: Complement L)

DXC

Double eXChange A and L (DXC L = CD: Complement A,L)

XCQ

eXChange Q (XCQ Q = CQ: Complement Q)

T

Test number in erasable/RAM

DT

Double Test double precision number in erasable/RAM

CP

ComPare A with number in erasable/RAM (CC as if c(A) – data word)

DCP

Double ComPare c(A,L) with DP number in erasable/RAM)

AD

ADd to A

DAD

Double precision ADd to A,L

SU

SUbtract from A

DSU

Double SUbtract from A,L

MP

MultiPly c(A) (DP product in A,L)

MSK

apply MaSK to A (logical AND)

DV

DiVide c(A,L) (quotient in A, remainder in L)

MPAD

MultiPly c(A) and ADd c(L) (DP product in A,L)

LAN

Load A Negative (LAN A = NA: Negate c(A))

DLN

Double Load Negative (DLN L = Negate c(A,L))

INC

INCrement by 1 or 2 per most recent precision

RED

REDuce by 1 or 2 as above; decrement & test loop count field

NDX

iNDeX next instruction by adding to its address field

STNA

STore Next Address to provide indirect address of next instruction
PRET*

distribute interPRETive instruction from ROM into A and L

Input/Output operations that address any of 512 channels; IR is 000; Q is 002
R

Read into A

ROR

Read & OR with A

RAND

Read & AND with A

RXOR

Read and XOR with A

W

Write from A

WOR

Write & OR with channel

WAND
Write & AND with channel

WXOR
Write & XOR with channel

Shifts in A,L or Q (negative shift count will reverse shift direction)
SDL

Shift Double Left (signed)

SDR

Shift Double Right (signed)

SDLU

Shift Double Left Unsigned

SDRU

Shift Double Right Unsigned

SAL

Shift A Left (signed)
SAR

Shift A Right (signed)

SALU

Shift A Left Unsigned

SARU

Shift A Right Unsigned

SLL

Shift L Left (signed)

SLR

Shift L Right (signed)

SLLU

Shift L Left Unsigned

SLRU

Shift L Right Unsigned

SQL

Shift Q Left (signed)

SQR

Shift Q Right (signed)

SQLU

Shift Q Left Unsigned

SQRU

Shift Q Right Unsigned

Miscellaneous operations with no address
PRA

Push Return Address onto return address list (converts to call type)
RUPT

Take all requested interRUPTs before proceeding

RSM

ReSuMe when interrupt is finished; restore saved CC

SAVCC
SAVe Condition Code

RSTCC
ReSTore Condition Code

REVIF

Edit condition code to REVerse sense of next IF

NORM
NORMalize c(A) and put shift count in Q
DNORM
Double NORMalize c(A,L) and put shift count in Q
ABA

ABsolute value of b(A) to A
ABL

ABsolute value in b(L) to L
DAB

ABsolute value in b(A,L) to A,L
ABQ

ABsolute value of b(Q) to Q

ROUND
Round c(A) per high bit of c(L)
SQRT

SQuare Root of c(A,L), root to A, remainder to L
SC

Sine of b(A) to A, Cos of b(A) to L
CPBK

CoPy BlocK of words (from-address in A, to-address in L, count in Q,


able to accommodate interrupts after any word copy). When preceded


by DATA=, copies the immediate data to the whole destination block.

Note on Condition Codes
In principle, the condition code can be represented in two bits, one for the sign and one for nonzero-ness, with a special case for overflow, which confers indeterminacy on those two conditions. But to accommodate the REVIF feature, it is actually kept as four bits, one to define the correct response by each of the IFs. REVIF translates the four-bit code to a different four-bit code to define the correct response under the reversed-if rule.
Cond
PNZ
NZ
NEG
NOV
Code/reversed N|Z
ZER
P|Z
OV
= 0
no
no
no
yes
0001/1110
yes
yes
yes
no

> 0
yes
yes
no
yes
1101/0010
no
no
yes
no

OVF
no
no
no
no
0000/0001
no
no
no
yes

< 0
no
yes
yes
yes
0111/1000
yes
no
no
no

The execution of any of these IF types notices the reversed state and if necessary translates back to a normal code. Also, while any reversed code is in force, no updates of it can be made, so any REVIF should by followed closely by the intended IF, either immediately or with at most an INDEX intervening. The code, in either state, can be saved in a dedicated special register by SAVCC and restored by RSTCC; this is done automatically when entering and leaving interrupt code (tasks).
Timing of the IF types depends on both the IF/ELSE decision. The decision is taken in the first pulse time: in the ELSE case, the one MCT fetches the next instruction per Z.
Sample Subroutine, Oriented More to Calculation than to Logic

P01 Double precision multiply, MPAC * memory, preserving max truncation
R02 triple precision. Call with little-end address in OperAddr.
 03 01234              DMP      PRA              CALL, not GOTO
 04 01235                       LDA    MPAC +1

 05 01236                       MP*    OperAddr  Product of low halves
 06 01237                       ROUND

 07 01240                       STA    L         Use only for precision
 08 01241                       LDA    MPAC

 09 01242                       MPAD*  OperAddr  First cross product
 10 01243                       DST    Temp      
 11 01244                       RED    OperAddr  Go to big-end address
 12 01245                       LDA    MPAC +1

 13 01246                       MP*    OperAddr  Second cross product
 14 01247                       DAD    Temp      Carryout goes to Q
 15 01250                       STL    MPAC +2

 16 01251                       STA    L         Upper cross prod sum
 17 01252                       LDA    MPAC

 18 01253                       MPAD*  OperAddr  Product of high halves
 19 01254                       AD     Q         Carryout (cross prods)
 20 01255                       DST    MPAC      
 21 01256                       RETURN           Assembles as GOTO 0
P50 Run code from erasable memory, using 2 words (op code and address)
R51 for each instruction. Emulates the native instructions without
R52 giving up control except where addressing high ROM requires it. 
R53 This “native-interpreter” (NI) task, whose priority is normally 
R54 zero, doesn’t run at all except when ground controllers load the
R55 code into RAM and then trigger a dedicated interrupt to change
R56 the task’s priority. Part of the ground load is the RAM address
R57 of the first instruction to be emulated, in location NISF.
R58 Runs in low ROM, hence the TC* types. Note: the condition code (CC)
R59 and its saved backup are indeterminate as the NI code begins. RUPT

R60 will not appear in such programs, being unnecessary since the NI
R61 itself performs RUPT frequently. Hence, no NI emulation of RUPT.

    RunFromE STQ    NIsaveQ   *RunFrmE is a permanent
             LDQ*   NISF
             STQ    NIopCode  Point to NI instruction
             INC    NISF      Point to address word

             LDQ*   NISF

             STQ    NIaddr    Usable as a direct address

             LDQ*   NIaddr    Don’t know if legit, but:

             STQ    NIindir    could be indirect address

             DATAQ  SignBit   Octal 40000

             XCQ    NISFback  Set unINDEXed for next inst

             IFNEG  +2        Skip restore when unINDEXed

             STQ    NISF      Restore saved NISF post NDX

          +2 LDQ*   NIopCode

             IFNEG  TransCtl  Any of 6 TC types: TransCtl
             DATAQ  IndirMsk  Octal 10000 [UPD FOR IPRET]
             MSK    NIaddr    Miscs always look direct ! Q NOT A !
             REVIF

             IFZER  Emulate   Assembles as +n because in E
    Indirect LDQ    NIindir

             STQ    NIaddr    Emulate indirect addressing  
             GOTO   Emulate
    TransCtl DATAQ  FonlyMsk  Octal 70000
             MSK    NIaddr    Get F-vs-E field, bits 15-13
             IFNZ
  MakeDir  F => direct in native mode
             DATAQ  FonlyMsk  Octal 70000
             MSK    NIindir   Get F-vs-E field, bits 15-13
             IFNZ   Emulate   F => indirect in native mode

             INC    NIopCode

             INC    NIopCode  NI modes follow native by +2
             GOTO   Emulate
    MakeDir  LDQ    NIaddr    In case INDEX changed E to F
             STQ    NIindir

    Emulate  INC    NISF
             LDQ    NISF
             STQ    NIZ       Both point to next opcode wd
             LDQ    NIsaveQ

             RSTCC           Note: overflow is a CC state
             EXEC*  NIopCode
             SAVCC           
             STQ    NIsaveQ   Because RUPT is a CALL if go BUT...
    NInext   IFIW   NIRupts   *NInext is perm, for TransCtl
             LDQ    NIsaveQ   In case another job had a go
             GOTO*  *RunFrmE  Until a native TC type escapes

P   Emulation routines for 15-bit address types. Note that all of them
R   except DATA= and DATAQ (which are after all just data operations)
R   begin with

R   a 15-bit address type providing a “sign” bit for the interpreter to

R   know which emulated codes can do transfers of control. The TC types
R   were separated into direct and indirect by the special assembler 

R   mode used by ground control. For each TC type, a one-word native
R   mode implementation is followed immediately by a one-word way to

R   resume erasable operation, then its NI mode logic.
    DATAQ    LDQ    NIaddr    No NI mode logic required
    +DATA    AD     NIaddr    No NI mode logic required

    CALL     CALL*  NIaddr

             GOTO*  *RunFrmE  In case of return from F

    NICALL   CALL   +1        Convert EXEC to CALL 

             PRA              Sets up return to EXEC +1



 LDQ    NIaddr

             XCQ    NIZ

             STQ    NIsaveQ

             LDQ    NIZ

             STQ    NISF

             INC    NIZ

             GOTO*  *NInext

    CALL*    CALL*  NIindir   INDEX may set = NIaddr

             GOTO*  *RunFrmE  In case of return from F

    NICALL*  CALL   +1        Convert EXEC to CALL

             PRA              Sets up return to EXEC +1

             LDQ    NIindir

             XCQ    NIZ

             STQ    NIsaveQ

             LDQ    NIZ

             STQ    NISF

             INC    NIZ

             GOTO*  *NInext

    EXEC     EXEC*  NIaddr

             GOTO*  *RunFrmE  In case of return from F

    NIEXEC   CALL   +1        Convert EXEC to CALL

             PRA              Sets up return to EXEC +1

             LDQ    NIaddr

             STQ    NISF      Notice no change in NIZ

             GOTO*  *NInext

    EXEC*    EXEC*  NIindir

             GOTO*  *RunFrmE  In case of return from F

    NIEXEC*  CALL   +1        Convert EXEC to CALL

             PRA              Sets up return to EXEC +1

             LDQ    NIindir

             STQ    NISF      Notice no change in NIZ

             GOTO*  *NInext

    IFPNZ    IFPNZ* NIaddr
             GOTO*  *RunFrmE  In case of return from F

    NIIFPNZ  IFPNZ* *NICAL1   Permanent -> NICALL+1

    IFPNZ*   IFPNZ* NIindir

             GOTO*  *RunFrmE  In case of return from F

    NIIFPNZ* IFPNZ* *NICAL*1  Permanent -> NICALL*+1
    IFNZ     IFNZ*  NIaddr

             GOTO*  *RunFrmE  In case of return from F

    NIIFNZ   IFNZ* *NICAL1    Permanent -> NICALL+1
    IFNZ*    IFNZ*  NIindir

             GOTO*  *RunFrmE  In case of return from F

    NIIFNZ*  IFNZ* *NICAL*1   Permanent -> NICALL*+1
    IFNEG    IFNEG* NIaddr

             GOTO*  *RunFrmE  In case of return from F

    NIIFNEG  IFNEG* *NICAL1   Permanent -> NICALL+1
    IFNEG*   IFNEG* NIindir

             GOTO*  *RunFrmE  In case of return from F

    NIIFNEG* IFNEG* *NICAL*1  Permanent -> NICALL*+1
    IFNOV    IFNOV* NIaddr

             GOTO*  *RunFrmE  In case of return from F

    NIIFNOV  IFNOV* *NICAL1   Permanent -> NICALL+1
    IFNOV*   IFNOV* NIindir

             GOTO*  *RunFrmE  In case of return from F

    NIIFNOV* IFNOV* *NICAL*1  Permanent -> NICALL*+1
P   Emulation routines for 12-bit address types. Note that all of them

R   begin with a non-15-bit address type providing a “sign” bit for the

R   interpreter to know which emulated codes cannot do transfers of

R   control. All are one-word native-mode implementations except INDEX.
    LDA      LDA*   NIaddr    Load A, set CC, REDL=1, OV=
    LDL      LDL*   NIaddr    Load L, set CC, REDL=1, OV=
    DLD      DLD*   NIaddr    Load A&L, set CC, REDL=2, OV=
    LDQ      LDQ*   NIaddr    Load Q, set CC, REDL=1, OV=
    STA      STA*   NIaddr    Store A, set CC, REDL=1, OV=
    STL      STL*   NIaddr    Store L, set CC, REDL=1, OV=
    DST      DST*   NIaddr    Store A&L, set CC, REDL=2, OV=
    STQ      STQ*   NIaddr    Store Q, set CC, REDL=1, OV=
    XCA      XCA*   NIaddr    Exchange A, set CC, REDL=1, OV=
    XCL      XCL*   NIaddr    Exchange L, set CC, REDL=1, OV=
    DXC      DXC*   NIaddr    Exchange A&L, set CC, REDL=2, OV=
    XCQ      XCQ*   NIaddr    Exchange Q, set CC, REDL=1, OV=
    LAN      LAN*   NIaddr    Load A negative, set CC, REDL=1, OV=

    DLD      DLD*   NIaddr    Load A&L negative, set CC, REDL=2, OV=

    CP       CP*    NIaddr    Compare single, set CC, REDL=1, OV=

    DCP      DCP*   NIaddr    Compare double, set CC, REDL=2, OV=

    AD       AD*    NIaddr    Add single, set CC, REDL=1, OV per

    DAD      DAD*   NIaddr    Add double, set CC, REDL=2, OV per

    SB       SB*    NIaddr    Subtract single, set CC, REDL=1, OV per

    DSB      DSB*   NIaddr    Subtract double, set CC, REDL=2, OV per
    MP       MP*    NIaddr    Multiply, set CC, REDL=1, OV=

    MPAD     MPAD*  NIaddr    Multiply & add, set CC, REDL=2, OV=

    DV       DV*    NIaddr    Divide, set CC, REDL=1, OV per

    ORQ      ORQ*   NIaddr    Or, set CC, REDL=1, OV=

    ANDQ     ANDQ*  NIaddr    And, set CC, REDL=1, OV=

    XORQ     XORQ*  NIaddr    Exclusive Or, set CC, REDL=1, OV=

    INC      INC*   NIaddr    Increment, set CC, REDL=1, OV per
    RED      RED*   NIaddr    Reduce per REDL, set CC, REDL=, OV per
    IPRET*   IPRET* NIindir   Load ROM word into A,L for interpreting
    INDEX    CALL   +1        Use of Z requires special NI logic

             PRA              Sets up return to EXEC +1

             DST    NIsaveAL  DP work weird but efficient
             LDL*   NIaddr    Amount to be added by INDEX
             DAD*   NIZ       L holds addr result ready to mask
             LDA*   NIZ       NIopCode-to-be

             STA    NIopNdxd
             DATAQ  LDA

             CP     Q
             IFNEG  Index15b  NI slot < LDA => DATAQ, DATA=, & TC types
             DATAQ  SHIFTS

             CP     Q

             IFNEG  Index12b  NI slot < SHIFTS => RAM addresses
             DATAQ  MISCELS
             CP     Q         NI slot < MISCELS => shifts & I/O

             REVIF

             IFP|Z  IndxDone  Indexing has no effect on MISCELS

    Index09b DATAQ  SHIOMsk   Octal 00777 for shifts and I/O

             GOTO   TrimIndx  Assembles as +4 because in E
    Index12b DATAQ  EMemMsk   Octal 07777 for RAM addresses
             GOTO   TrimIndx  Assembles as +2 because in E

    Index15b DATAQ  AllBits   Octal 77777 for ROM addresses

    TrimIndx STQ    AddrMsk
             ANDQ   L         Trim sum after index amt added

             XCQ    AddrMsk

             CQ               Mask for the part that stays put

             DLD*   NIZ       Address before indexing to L

             ANDQ   L

             ORQ    AddrMsk   Put whole address together

             STQ    NIadNdxd

             DATAQ  NIopNdxd

             XCQ    NISF
             STQ    NISFback

    IndxDone DLD    NIsaveAL
             RSTCC
             GOTO*  *NInext

Note on Sign Conventions in Double Precision

Double precision data expresses 29 bits in twos-complement notation, with the sign bit appearing in bit position 15 of the more significant word..

Support for Interpretive Instructions in ROM

The high-level interpretive instructions, 19 bits in length like the native instructions, have the following general format, effective only as an operand of a native IPRET instruction:

19 18 17 16 15 14 13 12 11 10 09 08 07 06 05 04 03 02 01

N  C  C  C  C  C  C  A  A  A  A  A  A  A  A  A  A  A  A

N=return to Native mode, C…C=operation Code, A…A= Address (RAM only).

Transition from native to interpretive mode is performed by a call to the interpreter, followed by interpretive instructions as in the Block II interpreter except for their being in the single-address format given above. The interpreter software saves the return address in an erasable location ZPret and performs IPRET* ZPret to copy the interpretive instruction into AL: N into A16 and A15, C…C into A6-01, and A…A into L12-01, setting other AL bits to zero. It increments ZPret, saves A in APret just to preserve the sign as a flag bit, and uses the AL data to invoke the appropriate routines to perform the interpretive instruction’s function. Then, if the saved APret is positive, it loops back to NextPret, or otherwise uses ZPret as a return address to resume native mode.
    Intrpret STQ    ZPret     Normal entry
    NextPret IPRET* ZPret     Set A & L per pret instruction
             INC    ZPret

             STA    APret     Negative if final pret instruc
    AnyPret  DATA=  PrOpMsk   Oct 00077
             MSK
             STQ    PretOpNo  (Might elim this if poss)
             STL    PretAddr  Used by most pret routines
             INDEX  PretOpNo

             GOTO   PretOpTb  Table of GOTOs to routines
    PostPret LDA    APret     Every routine end comes here
             IFNEG* ZPret     That was the final/only pret 

             GOTO   NextPret  Get and perform another

Sample Interpretive Routine
P01 Vector dot product, MPAC * memory, rounding result and clearing
R02 MPAC +2. Finds big-end vector address in PretAddr.

    VDotProd LDQ    PretAddr

             STQ    OperAddr

             INC    OperAddr  Point to little end of X component

             CALL   DMP

R            DLD    MPAC      C(AL) = C(MPAC) anyway
             DST    Tempac

             ZA

             LDL    MPAC +2

             DST    Tempac +2 Initialize temp accumulator

             DLD    MPAC +3
             DST    MPAC

             INC    OperAddr

             INC    OperAddr  Point to little end of Y component

             CALL   DMP

R            DLD    MPAC      C(AL) = C(MPAC) anyway

             DAD    Tempac
             DST    Tempac

             ZA

             LDL    MPAC +2

             DAD    Tempac +2

             DST    Tempac +2 Build rounding quantity

             DLD    MPAC +5

             DST    MPAC

             INC    OperAddr

             INC    OperAddr  Point to little end of Z component

             CALL   DMP

R            DLD    MPAC      C(AL) = C(MPAC) anyway

             DAD    Tempac

             DST    MPAC
             DZ
             XCL    MPAC +2   Clear MPAC +2 as specified
             DAD    Tempac +2 Complete rounding quantity in A
             STA    L

             ZA

             DAD    MPAC

             DST    MPAC      Round

             GOTO   PostPret

Survey of Interpretive Instructions

The guiding principle for the Block III Interpreter is to have exactly the same functional capabilities as the Block II Interpreter but with different instruction formats and the wider access to memory implied by the Block III architecture.

Loading Data into MPAC and Setting MODE; Scalar Arithmetic; Vector Arithmetic

00-03 SLOAD, DLOAD, TLOAD, VLOAD
04-14 IDAD, IDSB, BDSB, DMP, DMPR, DDV, BDDV, SIGN, TAD

15-26 VAD, VSB, BVSB, DOT, VXSC, V/SC, VXV, VPROJ, VXM, MXV

Storing Data from MPAC, according to prior state of MODE; Store SP constant
27-33 STORE, STOX1, STOX2, SSP, STQ [STODL, STOVL, STCALL not req’d]
Unary Scalar Arithmetic; Unary Vector Arithmetic (up to 4K, in principle!)
34 PUSH, SQRT, SIN, COS, ASIN, ACOS, DSQ, ROUND, DCHS, ABS
   UNIT, ABVAL, VSQ, VCHS, VDEF, RVQ
Scalar Shifts (NORM stores a result); Vector Shifts

35-37 NORM, SR, SL, SRR, SLR, VSR, VSL
Unconditional Transfer of Control

41-42 IGOTO, ICALL
43    RTNM [EXIT not required since built in to all codes]
Branching

44-53 BPL, BZE, BMN, BHIZ, BOV, BOVNM, BIX1, BIX2
Index register processing

54-57 X1=, X2=, X1+=, X2+=

60-63 LX1, LX2, LX1-, LX2-
64-67 STX1, STX2, XCX1, XCX2
70-73 X1AD, X2AD, X1SB, X2SB
Discrete Processing using any of 60 discretes (2 bits Br, 2 bits Op, 8 bits No)
74 NOP, SET, CLEAR, INV

   GO2, SETGO, CLRGO, INVGO

   BON, BNSET, BNCLR, BNINV

   BOFF, BFSET, BFCLR, BFINV

Pushdown List Processing

75-77 SETPD, PDDL, PDVL

Note on Interpretive addressing: all instructions that read word data treat addresses below 0100 specially, using the job’s work area as a base register for values below 0074, taking 0077 as a pop command, and taking 0074, 0075 and 0076 to mean indexing, subtracting nothing, X1 or X2 respectively from a full 15-bit address to access erasable or high fixed.
Storing instructions use the work area for addresses below 0074, but do nothing for addresses 0074-0077.

Need to work on maybe special SETPD too.
Sample Pulse Sequences
In all but a few special cases, Time 6 (the final microsecond of each MCT) is implicit, either using staging code bits to switch to the next MCT of the current instruction, or copying the next instruction from the fixed-memory local register GF into the sequence control register SQ and the general buffer register B (control pulse F2SQB). Stage 0 is a universal prologue for indirect addressing. When there is no indirect addressing, the sequence begins with Stage 1.
INDEX1 – Index next instruction
1. RUXB WSE Z2SF         // Address from U if indexed, else from B.
2.

3. RE WY                 // Amount by which to index next instr.
4.

5. INCZ NISQ             // Standard proceed-to-next-instruction.
6. F2SQBX NDXT           // Special: add and flag, blocking *INCs.
DATAA1 – Place immediate data into A; set CC per A
1. RUXB WA WCC Z2SF     // Address field to A and Condition.
2.                      // Note: indexing DATAA may be useful.
3. INCZ NISQ

INDIR0 - Indirect-address prologue. Setting BR1 & BR2
1. RUXB WSE             // Yes, indexed indirect may be useful.
2. RE WY DPCI           // For double precision instructions only,
3. RU WB                // convert big-endian to little-endian.
4. BR1 BR2 ST1          // 2 BRs block central register addressing.
IPRET1 – Fetch one interpretive instruction from ROM

1. RUXB WSF
2. ST2

IPRET2 – Set up AL for pret & fetch next native instruction

1. F2ALCC Z2SF           // F19→A16, F18-13→A6-1, F12-1→L12-1.
2. INCZ NISQ

EXEC1 – Perform one native instruction from ROM

1. RUXB WSF

2. NISQ                 // Secret of EXEC is leaving Z unchanged.
CALL1 – Call program module in ROM

1. RUXB WSF

2. RZ WQ                // Functions as a CALL if STQ at beginning
3. RB WZ                // of module; else functions as a GOTO.
4. INCZ NISQ

IFPNZ1 – Branch on CC Positive Non-Zero (REVIF => IFN|Z)
1. C=1xxx RUXB WSF       // Use effective address if branching,
1. C=0xxx Z2SF           // else ignore address and carry on.
2. C=1xxx RZ WQ          // All IF types set Q like CALL.
3. C=1xxx RB WZ
4.
5. INCZ NISQ

IFNZ1 – Branch on CC Non-Zero (REVIF => IFZER)

1. C=x1xx RUXB WSF

1. C=x0xx Z2SF

2. C=x1xx RZ WQ

3. C=x1xx RB WZ
4. 
5. INCZ NISQ

IFNEG1 – Branch on CC Negative (REVIF => IFP|Z)

1. C=xx1x RUXB WSF

1. C=xx0x Z2SF

2. C=xx1x RZ WQ

3. C=xx1x RB WZ
4. 
5. INCZ NISQ

IFNOV1 – Branch on CC No OVerflow (REVIF => IFOV)

1. C=xxx1 RUXB WSF

1. C=xxx0 Z2SF

2. C=xxx1 RZ WQ

3. C=xxx1 RB WZ
4. 
5. INCZ NISQ

RUPT1 – Take highest priority interrupt [TO BE REWRITTEN]
1. IW  RUXB WSF

1. NIW Z2SF

2. IW  RZ WQ

3. IW  RB WZ

4. 

5. INCZ NISQ

LDA1 – Load A and set CC
1. RUXB WSE Z2SF        
2. B=00 RA WCC          // LDA A = TA: just test it.

2. B=01 RL WA WCC       // LDA L & LDA Q work as expected.

2. B=10 RQ WA WCC
3. B=11 RE WA WCC       // Indexed, indirect, or addr > 2.
4.

5. INCZ NISQ

LDL1 – Load L and set CC
1. RUXB WSE Z2SF        

2. B=00 RA WL WCC       // LDL A & LDL Q work as expected.

2. B=01 RL WCC          // LDL L = TL: just test it.

2. B=10 RQ WL WCC

3. B=11 RE WL WCC       // Indexed, indirect, or addr > 2.

4.

5. INCZ NISQ

DLD1 – Double Load, L part or DT if direct address < 3
1. B=11 RUXB WSE WQ

1. B<11 Z2SF

2. B=11 RQ WY MONEX
2. B<11 RL WCC          // Any address 0-2 treated as DLD A.
3. B=11 RE WL WCC

3. B<11 RA WCCD         // DLD A = DT: Test DP value in AL.
4.                      // Assumes C(A,L) is sign-aligned.
5. B=11 RU WB ST2
5. B<11 INCZ NISQ

DLD2 – Double Load, A part and set CC for the combination
1. RB WSE Z2SF
2. 

3. RE WY WCCD TSDAL     // May have to sign-align new data.
4. ADJA

5. RU WA INCZ NISQ

LDQ1 – Load Q and set CC
1. RUXB WSE Z2SF        

2. B=00 RA WQ WCC       // LDQ A & LDQ L work as expected.

2. B=01 RL WQ WCC

2. B=10 RQ WCC          // LDQ Q = TQ: just test it.

3. B=11 RE WL WCC       // Indexed, indirect, or addr > 2.

4.

5. INCZ NISQ

R Lots of problems here with (dir) CR addressing and special cases
STA1 – Store A, or Zero A if direct address = 0
1. RUXB WSE Z2SF        

2. B=00 WA WCC          // STA A = ZA: just zero it.

2. B=01 RA WL WCC       // STA L & STA Q work as expected.

2. B=10 RA WQ WCC

3.

4. B=11 RA WE WCC       // Indexed, indirect, or addr > 2.

5. INCZ NISQ

STL1 – Store L, or Zero L if direct address = 1
1. RUXB WSE Z2SF        

2. B=00 RL WA WCC       // STL A & STL Q work as expected.

2. B=01 WL WCC          // STL L = ZL: just zero it.

2. B=10 RL WQ WCC

3. 

4. B=11 RQ WE WCC       // Indexed, indirect, or addr > 2.

5. INCZ NISQ

DST1 – Double Store, L part; DZ case when dir addr < 3
1. B=11 RUXB WSE WQ     // Borrow Q as buffer to get upper.
1. B<11 Z2SF            // No ST2 cycle req’d when doing DZ.
2. B=11 RQ WY MONEX TSDAL // Signs of A, L set INVSL, ADJA
2. B<11 WA WL WCC       // Any address 0-2 treated as ZD.

3. B=11 INVSL
4. B=11 RL WE WCC       // Store sign-aligned L this cycle.
5. B=11 RU WB ST2
5. B<11 INCZ NISQ

DST2 – Double Store, A part
1. Z2SF RB WSE

2. RA WY ADJA           // ±1, or 0, to X according to TSDAL.
3. 

4. RU WA WE WCCD        // Keep sign-aligned C(A,L).
5. INCZ NISQ

STQ1 – Store Q, or Zero Q if direct address = 2
1. RUXB WSE Z2SF        

2. B=00 RQ WA WCC       // STQ A & STQ L work as expected.

2. B=01 RQ WL WCC

2. B=10 WQ WCC          // STQ Q = ZQ: just zero it.

3.

4. B=11 RQ WE WCC       // Indexed, indirect, or addr > 2.

5. INCZ NISQ

XCA1 – Exchange A

1. Z2SF RB WSE

2. 0xx RA WY TCR 

2. 1xx RA WY 

3. RGE WA WCC

4. RU WGE

5. INCZ NISQ

XCL1 – Exchange L
1. Z2SF RB WSE

2. RL WB

3. RGE WL WCC

4. RB WGE

5. INCZ NISQ

DXC1 – Double Exchange, L part
1. RB WSE WY MONEX TSDAL
2. 0x RL WB

2. 1x RL WB INVSL
3. RGE WL WCC
4. RB WGE
5. RU WB ST2

DXC2 – Double Exchange, A part
1. Z2SF RB WSE

2. 0x RA WY

2. 10 MONEX RA WY

2. 11 PONEX RA WY

3. RGE WA WCCD 
4. RU WGE
5. INCZ NISQ

XCQ1 – Exchange Q
1. Z2SF RB WSE

2. RQ WB
3. RGE WQ WCC
4. RB WGE 

5. INCZ NISQ

Logic for NORM: Normalize Signed Double Precision Number in A and L
NORM shifts C(AL) numerically until it is normalized with no overflow present (A16=A15≠A14), even if overflow is initially present, and places in Q the shift count required to do this. If C(AL) is zero, no shifting occurs, the zero condition is set, Q = 0.

The duration is 6 PT = 1 MCT if C(AL) is initially normalized or nearly so. Otherwise each shift beyond the first two takes 2 additional PT. Like all shift instructions, this one can insert a sub-MCT gap into the succession of MCTs.
Fetching next instruction begins (Z2SF) unconditionally in PT1. Also in PT1, Q = -1 and C(AL) shifted right 1 copies to ALS, CC reflects AL.

In PT2, if CC/OV then C(ALS) copies to AL and NISQ, else adder inputs X & Y are zeroed. This occurrence of Next Instruction to SQ (NISQ) has a delayed effect (to PT6) because a ROM cycle is in progress; in the intervening pulse times no actions occur.
In PT3, C(AL) shifted left 1 copies to ALS, adder sum U goes to Q (value zero in first occurrence of PT3), CC reflects new ALS.

In PT4, if CC/OV then CC reflects AL, NISQ (delayed if first PT4), else C(ALS) copies unshifted to AL, Q enters adder to be incremented, force PT3 as next step.

No pulses are specified for PT5, but when PT4 takes the else branch, PT3 repeats: C(AL) shifted left 1 copies to ALS, adder sum U goes to Q (an increase of 1 in non-first PT3), CC reflects new ALS.
No pulses are specified for PT6, but when PT3 logic was repeated in occurred in PT5, PT4 repeats: if CC/OV then CC reflects AL, NISQ (here not delayed, because the ROM cycle is complete), else C(ALS) copies unshifted to AL, Q enters the adder to be incremented, force PT3 as the next step.
The logic of PT3 and PT4 repeats until normalization is achieved, and the first NISQ is effective at once. Thus, the duration of NORM, and the shift count in Q, can be: 

  6 PT when Q = -1, 0, or 1 (1 MCT)
  8 PT when Q = 2

10 PT when Q = 3

12 PT when Q = 4 (2 MCT)
. . . . . . . . . . . . . . . 

60 PT when Q = 28 (10 MCT)
61 PT when Q = 29

That maximum may be be the longest instruction duration in Block III, but it’s not seriously out of line. Statistically, 1 or <2 MCT will be predominant, especially when normalizing data obtained from the Interrupt Request (IR) register while entering an interrupt. NORM plays a key role in identifying the highest-priority interrupt request.
Instruction Sequencing Logic, Including IFs, RUPT, and Cycle Steals
Most instructions, most of the time, will be completed in one Memory Cycle Time, specifically those whose data-handling functions can be completed in one cycle of RAM/erasable. In these cases, Pulse Time 1 copies the data address to SE and the next-instruction address (usually from Z) to SF.
In all cases, pulse NISQ (Next Instruction to SQ) establishes the current memory cycle as the last one in the current instruction. Usually, it is coded in Pulse Time 6, where it advances the operation code to the diode-matrix SeQuence generator where it will become effective in the following PT1. However, other pulses coded into PT6 of the current instruction will function. In some cases, like NORM when 2 or more shifts are needed, or other shift instructions when the number of shifts exceeds the capacity of one MCT, NISQ may appear in earlier Pulse Times. Those NISQs are effective immediately when no ROM or RAM cycle is in progress, but if a memory cycle is in progress, the effect is held until PT6 of that cycle; moreover, any pulses coded into Pulse Times after the NISQ but in or preceding PT6 are blocked and ineffective.
When a NISQ pulse becomes effective, what it puts into SQ can be overridden by the highest-priority cycle-steal request, so that the “stolen” RAM cycle occurs between instructions and can use data-handling parts of the ALU. For that reason, there is an exception: if any inputs to X or Y have been done after the most recent RU pulse, such cycle steals are blocked because the adder output U will be needed in the next instruction.
Whenever a ROM/fixed cycle begins, i.e. when an address is written into SF, dedicated incrementer logic begins, so that pulse RSF (Read SF) will collect the next higher address for other uses. Tentatively, RSF will be available from PT3 onward in that cycle, to allow a simple incrementer without explicit-carry logic. Hopefully, PT2 will turn out to work equally well. Usually, the address read by RSF will be written into Z, setting it up for use in the next cycle.
Any IF-type instruction, when its PT1 finds the stated condition unsatisfied, operates as a NOP, copying Z to SF and the incremented version back to Z like any ordinary instruction.
Actual transfers of control (except for EXEC and RUPT) work a little differently. PT1 copies the instruction’s address field into SF, also copying Z into Q as a possible return address before RSF updates Z. Because any transfer of control may turn out to be a CALL rather than a GOTO, the Return Address Pointer, RAP, is placed into Y and a carry-in ordered to increment it, but the sum is not read out (by RU) in this cycle. Instead, the sum is carried over into the first cycle of the instruction to which control is being transferred, blocking intervening cycle steals as mentioned above. We’ll get back to what happens there in a minute.
EXEC also copies its address field into SF in PT1 to start the fetching of the instruction being EXECuted, but places that address’s incremented form into LS instead of Z. By leaving Z unchanged, it sets up the instruction after the EXEC to be the successor to the EXECuted instruction (unless it performs a successful transfer of control). Then it performs the same “unfinished business” logic on RAP as in any other transfer of control, blocking cycle steal(s) between the EXEC and the instruction being EXECuted. Note that EXEC passes more data to its target instruction than other transfers: Q as well as A and L.
RUPT is a special transfer of control that needs no address field because it can transfer control only to ROM location 00000. If its PT1 finds IR to be all zero, the copying of Z to SF and putting the incremented value into Z occurs as in ordinary instructions, and no transfer of control takes place. But when PT1 sees any non-zero bits in IR, it zeros both SE and SF, copies Z into adder input Y, and starts a decrement. The 0000 in SE addresses QRUPT, into which Q is saved, after which the sum, b(Z)-1, goes into Q to establish the address of the RUPT itself as the resume address. Then it performs the same “unfinished business” logic on RAP as in any other transfaer of control, blocking cycle steal(s) between the RUPT and the instruction (mandated to be a PRA) in 00000.
Almost all of this RUPT logic is shared with the Copy Block (CPBK) instruction, using the same test of IR at the beginning of each of its read cycles to switch its action to the RUPT sequence given here. When IR is found to be zero, CPBK’s two RAM cycles (read from source and write into destination) continue normally, incrementing the address pointers in A and L, and decrementing and testing Q to see whether the block transfer is finished. This allows interrupts to be taken at any point in the block transfer, such that resuming after each interrupt executes the CPBK again, but with its counters set to pick up where it left off. That works because Q is saved in QRUPT as described above, and because A and L are saved at the interrupt entry point (by mandate) and restored just before the resume. All 3 of those built-in register saves are peculiar to RUPT, and the A and L restores are not done in regular call-return sequences. The reason the RUPT logic is parallelled with the read cycle and not the write cycle is that a CPBK preceded by a DATA= code omits read cycles and just puts the immediate data into all the destination locations without allowing any interrupts. Letting it allow such interrupts would require more logic to retain the immediate-data state through each interrupt.
The question of whether a transfer of control is a CALL or a GOTO is decided at the entry-point instruction receiving control. If a CALL is intended, the entry-point instruction itself must be Push Return Address (PRA), a no-address type. In PT1, PRA copies the adder sum U into RAP (completing the extension of the RA list by one word) and into SE (beginning the cycle that fills the new head of the RA list)—also finishing the “unfinished business” from the transfer-control instruction. Then it copies Q to the RAM location thus addressed, populating the new head. It also performs SAVCC.
The ReturnAddress (RA) list occupies RAM locations 0001 up to (at most) 0077. Thus a 0000 in RAP indicates an empty list, ready for the increment that occurs when an RA is pushed onto the list. Location 0000 itself is dedicated to interrupt storage QRUPT.

What follows the PRA is whatever processing the subroutine has to do, including any CALLs to lower-level subroutines it requires. Returning from the subroutine can be done from any number of places within it, by any successful transfer of control instruction (conditional or unconditional) with an address of 00000. The all-zero address field is recognized as a special case to invoke the Return logic, which starts by copying RAP to SE and to Y for decrementing. Then it puts the adder result into RAP, popping the head off the Return-Address list. The address read from RAM goes into Z, and the next cycle 
returns control to the calling routine. It copies Z into SF to start the read, copies the incremented version back into Z in the usual way, and copies the permanent QRUPT address (0000) into SE and thus restores Q as required for resuming after an interrupt, though that is not mandated for other returns. However, it is beneficial if the original call was preceded by STQ QRUPT, which is legitimate in any non-task code. Note also that returning from code invoked by RUPT or CPBK (i.e. the RSM instruction as shown below) gives control back to the same RUPT or CPBK so that it can again check IR for Interrupt Requests and invoke the highest-priority of those before continuing the current Job. RSM differs from GOTO 00000 only in adding restoration of the saved Condition Code.
The entry point targeted by EXEC will usually be a one-instruction subroutine, often enough a DATA= or DATAA, simply using the undisturbed state of Z to return control to the instruction following the EXEC. Usually, the EXEC will be INDEXed. Sometimes one or a few of the possible targets for a given EXEC will be a routine that is not complete in one instruction, in which case it must be a CALL, as seen below.
Interrupt Processing

P01 Code, located starting at 00000, common to all interrupts. Takes
R02 15 MCT plus a little for possibly longer shifts, plus RuptX time.
 03 00000              RUPT     PRA              Convert to CALL; SAVCC
 04 00001                       DST    ALRupt    Save A,L in case CPBK

 05 00002                       READ   IR        Still has seen request
 06 00003                       NORM             Find highest prio req
 07 00004                       DATAA  Rupt0Com  57777 as is for RUPT 0

 08 00005                       NDX    Q  

 09 00006                       SAR    0         Note sign extension
 10 00007                       WAND   IR        Turn off RUPT request
 11 00010                       NDX    Q
 12 00011                       EXEC   RuptVect  NDX picks out a CALL
 13 00012                       DLD    ALRupt    Restore in case CPBK
 14 00013                       RSM              GOTO 0 plus RSTCC
 15 00014              RuptVect CALL  Rupt0

 16 00015                       CALL  Rupt1

 17 00016                       CALL  Rupt2

. . . . . . . . . . . . . . . . . . . . . . .

 nn 000xx                       CALL  Ruptn
Change-of-Job Checking

                                T      NewJob    NZ when a higher pri-

                                IFPNZ  ChangJob   ority job is waiting

